

Formulation and Analysis of a Dynamic Spatial Price Equilibrium Model

Thomas F. Rutherford
Economics 6433

March, 2002

1 The Static Model

The spatial price equilibrium model begins with the Hitchcock-Koopmans transportation problem as described by Dantzig (1963). The data for this model includes a set of suppliers I and a set of markets J , with supplies a_i , demands b_j , transport costs c_{ij} from supplier i to market j . Cast as a planning problem, this linear program seeks a transport schedule which minimizes the cost of supplying all markets. That is:

$$\begin{aligned} \min \quad & \sum_{ij} c_{ij} x_{ij} \\ \text{s.t.} \quad & \sum_j x_{ij} \leq a_i \\ & \sum_i x_{ij} \geq b_j \\ & x \geq 0 \end{aligned}$$

Table 1 presents GAMS code defining the sets and data parameters, and Table 2 presents GAMS code for the linear programming formulation of this model.

It is well known that this problem can be interpreted as a market equilibrium problem in which the dual multiplier for an equation from the first set of constraints, w_i , represents the price in supply market i , and the dual multiplier for an equation in the second set of constraints, p_j , represents the price in demand market j .

Cast as a market equilibrium problem, a system of IJ inequalities and associated complementary slackness conditions replace the minimization operator. “Zero profit conditions” for transportation activities characterize a competitive, constant-returns-to-scale (CRTS) “transportation” industry in which free entry drives excess profits to zero for active trade links and no unprofitable activities are operated. The equations which char-

```

SETS
  I   canning plants   / SEATTLE, SAN-DIEGO /
  J   markets          / NEW-YORK, CHICAGO, TOPEKA / ;

PARAMETERS
  A(I) capacity of plant i in cases (when prices are unity)
    /   SEATTLE      325
      SAN-DIEGO     575 /,
  B(J) demand at market j in cases (when prices equal unity)
    /   NEW-YORK     325
      CHICAGO        300
      TOPEKA         275 /,

  ETA(I) Price elasticity of supply
    /   SEATTLE      1.0
      SAN-DIEGO     1.0 /,

  SIGMA(J) Price elasticity of demand
    /   NEW-YORK     1.5
      CHICAGO        1.2
      TOPEKA         2.0 /;

TABLE DIST(I,J) distance in thousands of miles
      NEW-YORK    CHICAGO    TOPEKA
SEATTLE      2.5      1.7      1.8
SAN-DIEGO    2.5      1.8      1.4 ;

SCALAR F freight in dollars per case per thousand miles /90/ ;

PARAMETER C(I,J) transport cost in thousands of dollars per case ;

C(I,J) = F * DIST(I,J) / 1000 ;

PARAMETER PBAR(J) Reference price at demand node J (supply price = 1)
    /   NEW-YORK     1.225
      CHICAGO        1.153
      TOPEKA         1.126 /;

```

Table 1: GAMS Set and Data Statements for the Transportation Model

```

VARIABLES
  X(I,J) SHIPMENT QUANTITY FROM I TO J
  COST   MINIMAND - TOTAL COST OF SHIPMENT

POSITIVE VARIABLE X;

EQUATIONS
  SUPPLY(I) SUPPLY LIMIT
  DEMAND(J) DEMAND CONSTRAINT (FIXED)
  OBJDEF    DEFINES COST;

SUPPLY(I).. A(I) =G= SUM(J, X(I,J));
DEMAND(J).. SUM(I, X(I,J)) =G= B(J);
OBJDEF..    COST =E= SUM((I,J), C(I,J) * X(I,J));

MODEL MINCOST / SUPPLY, DEMAND, OBJDEF/;

SOLVE MINCOST USING LP MINIMIZING COST;

```

Table 2: The Transportation Model Formulated as a Linear Program

```

POSITIVE VARIABLES
    W(I)          SHADOW PRICE AT SUPPLY NODE I,
    P(J)          SHADOW PRICE AT DEMAND NODE J,
    X(I,J)        SHIPMENT QUANTITIES IN CASES;

EQUATIONS
    SUPPLY(I)      SUPPLY LIMIT AT PLANT I,
    DEMAND(J)      FIXED DEMAND AT MARKET J,
    PROFIT(I,J)    ZERO PROFIT CONDITIONS;

PROFIT(I,J)..     W(I) + C(I,J)   =G= P(J);

SUPPLY(I)..       A(I) =G= SUM(J, X(I,J));

DEMAND(J)..       SUM(I, X(I,J)) =G= B(J);

MODEL TRNSP / PROFIT.X, SUPPLY.W, DEMAND.P / ;

SOLVE TRNSP USING MCP;

```

Table 3: The Transportation LP Formulated as an LCP

acterize this equilibrium are:

$$\sum_j x_{ij} \leq a_i, \quad w_i \geq 0, \quad w_i (a_i - \sum_j x_{ij}) = 0 \quad \forall i$$

$$\sum_i x_{ij} \geq b_j, \quad p_j \geq 0, \quad p_j (b_j - \sum_i x_{ij}) = 0 \quad \forall j$$

$$w_i + c_{ij} \geq p_j, \quad x_{ij} \geq 0, \quad x_{ij} (w_i + c_{ij} - p_j) = 0 \quad \forall i, j$$

This system of inequalities is a *linear* complementarity problem which may be specified in GAMS/MCP syntax as shown in Table 3.

There would be no particular reason to formulate a linear program as an MCP. Suppose, however, that demands and supplies were price responsive and all markets perfectly competitive. These assumptions alone would not rule out an optimization approach, but they would require use of a *nonlinear* programming algorithm. For concreteness, let demand and supply functions be isoelastic. The revised equilibrium conditions are:

$$\sum_j x_{ij} \leq \alpha_i w_i^{\eta_i}, \quad w_i \geq 0, \quad w_i (\alpha_i w_i^{\eta_i} - \sum_j x_{ij}) = 0 \quad \forall i$$

$$\sum_i x_{ij} \geq \beta_j p_j^{-\sigma_j}, \quad p_j \geq 0, \quad p_j (\beta_j p_j^{-\sigma_j} - \sum_i x_{ij}) = 0 \quad \forall j$$

$$w_i + c_{ij} \geq p_j, \quad x_{ij} \geq 0, \quad x_{ij} (w_i + c_{ij} - p_j) = 0 \quad \forall i, j$$

This is a nonlinear complementarity problem which in certain cases may represent the first-order optimality conditions for a nonlinear program. When the matrix of cross-price elasticities for a partial equilibrium model is symmetric, there is an associated optimization problem which can be used to compute the equilibrium prices and quantities. (The model is said to be *integrable* - see Takayama and Judge (1971).) In the present example,

```

$STITLE  A SPATIAL PRICE EQUILIBRIUM MODEL FORMULATED AS AN NLP

VARIABLES      X(I,J)      SHIPMENT QUANTITY FROM I TO J
                Y(I)        SUPPLY
                D(J)        DEMAND
                SURPLUS      MAXIMAND - SOCIAL SURPLUS;

POSITIVE VARIABLE X, Y, D;

EQUATIONS      SUPPLY(I)    SUPPLY LIMIT
                DEMAND(J)    DEMAND CONSTRAINT (FIXED)
                OBJDEF       DEFINES COST;

SUPPLY(I)..    Y(I) =G= SUM(J, X(I,J));

DEMAND(J)..    SUM(I, X(I,J)) =G= D(J);

OBJDEF..       SURPLUS =E=  SUM(J, KAPPA(J) * D(J)**EPSD(J))
                           - SUM(I, GAMMA(I) * Y(I)**EPSY(I))
                           - SUM((I,J), C(I,J) * X(I,J));

MODEL NLPEQUIL /SUPPLY, DEMAND, OBJDEF/;

D.LO(J) = 0.01;

SOLVE NLPEQUIL USING NLP MAXIMIZING SURPLUS;

```

Table 4: The Spatial Price Equilibrium Model in an Optimization Format

cross-price elasticities are zero, and the associated optimization problem is:

$$\begin{aligned}
& \max \sum_i \gamma_i y_i^{\epsilon_i^y} + \sum_j \kappa_j d_j^{\epsilon_j^d} - \sum_{ij} c_{ij} x_{ij} \\
& \text{s.t.} \quad \sum_j x_{ij} \leq y_i \\
& \quad \quad \sum_i x_{ij} \geq d_j \\
& \quad \quad x_{ij} \geq 0, \quad d_j \geq 0, \quad y_i \geq 0
\end{aligned}$$

where:

$$\gamma_i = \frac{\eta_i}{1 + \eta_i} \left(\frac{1}{\alpha_i} \right)^{1/\eta_i}, \quad \kappa_j = \frac{\sigma_j}{1 + \sigma_j} \beta_j^{1/\sigma_j}, \quad \epsilon_i^y = \frac{1 + \eta_i}{\eta_i}, \quad \text{and} \quad \epsilon_j^d = \frac{\sigma_j - 1}{\sigma_j}$$

In economic terms, the market allocation “maximizes the sum of producer and consumer surplus”.

Economic equilibrium models are typically used to assess the consequences of market distortions, typically in the form of taxes, tariffs or other types of government policies. In public finance, equilibrium models are often used to measure the “excess burden” of a given tax structure. When *ad-valorem* taxes are present, the NLP formulation of a market equilibrium problem is not straight-forward. No single optimization problem characterizes the equilibrium because the resulting allocation is inefficient. Such an equilibrium could be computed by solving a *sequence* of nonlinear programs, but in these cases the MCP formulation is certainly more transparent.

```

PARAMETER      T(I,J)      AD-VALOREM TAX RATE;

POSITIVE VARIABLES
      W(I)      SHADOW PRICE AT SUPPLY NODE I,
      P(J)      SHADOW PRICE AT DEMAND NODE J,
      X(I,J)    SHIPMENT QUANTITIES IN CASES;

EQUATIONS
      SUPPLY(I)    SUPPLY LIMIT
      DEMAND(J)    DEMAND CONSTRAINT (FIXED)
      PROFIT(I,J)  ZERO PROFIT CONDITIONS;

SUPPLY(I)..      ALPHA(I) * W(I)**ETA(I) =G= SUM(J, X(I,J));
DEMAND(J)..      SUM(I, X(I,J)) =G= BETA(J) * P(J)**(-SIGMA(J));
PROFIT(I,J)..    (1 + T(I,J)) * (W(I) + C(I,J)) =G= P(J);

MODEL TRNSP / PROFIT.X, SUPPLY.W, DEMAND.P/ ;

*      Avoid function evaluation errors by installing positive bounds:
P.LO(J) = 0.001;  W.LO(I) = 0.001;

*      10% tax on all trade:
T(I,J) = 0.10;

SOLVE TRNSP USING MCP;

```

Table 5: Spatial Price Equilibrium with Ad-Valorem Taxes

As illustration of a model which is difficult to treat in an optimization framework but easy to pose as an MCP, consider applying *ad-valorem* taxes on trade flows in the spatial equilibrium model from above. This causes the supply price to be a non-unitary multiple of the marginal cost of supply, destroying integrability (notice that *specific* taxes do not cause this problem, as they can simply be added to the transport cost coefficients). The GAMS/MCP model is shown in Table 4.

2 Calibration

We begin from a benchmark dataset describing supply, demand and transport cost and trade barriers for a single commodity. The most readily available sources of these data would be the U.S. Department of Energy's Energy Information Agency (see <http://www.eia.doe.gov/>) and the U.S. Department of Agriculture Economic Research Service (see <http://www.ers.usda.gov/>). We want to build a spatial price equilibrium model for a single homogeneous commodity. A minimal dataset for this work would include:

\bar{S}_i base year supply from region i

\bar{D}_j base year demand in region j

c_{ij} transport cost from region i to region j

t_{ij} specific unit tax on sales from region i to region j

The tax rates specified as part of the benchmark data are quite general. They could be used to represent a combination of production subsidies (s_i), production taxes (T_i) and import tariffs (τ_{ij}) by making the assignment:

$$t_{ij} = T_i - s_i + \tau_{ij}$$

In representing the base year equilibrium the individual components of t_{ij} are irrelevant, but in subsequent policy analysis, the distributional consequences of reforms will depend on how tax revenues accrue to various regional governments.

The following GAMS code provides data for a spatial price equilibrium dataset. The table has been imported from an Excel worksheet and pasted into the GAMS program file using a conventional text editor:

```
$title Wheat Production and Consumption data

set      R           Regions /
        USA         United States
        CAN         Canada
        ...
        /;

set      col         Data columns /Area,Yield,Produce,Consume/
        yr           Years /1996*2000/;

Table data(r,yr,col) Wheat Market Data (1000s)
        Area  Yield  Produce  Consume  Price
can.1996  12262.00  2.43  29801.00  8222.00  12.0
can.1997  11410.00  2.13  24280.00  7336.00  10.2
can.1998  10769.00  2.24  24076.00  8077.00  8.9
can.1999  10364.00  2.59  26850.00  7900.00  9.5
can.2000  10900.00  2.46  26800.00  8200.00  10.1
...

Table trncost(r,r) Transport costs (% of CIF value)
        usa    can    jpn    gbr    deu    den    ...
can      2      0      5      8      9      2
...
```

Let the GAMS variable named `x0` to represent the base year data which is denoted \bar{x} here. When we read in the base year data, it is convenient to *impose* market consistency in order to have a reference equilibrium point. Here I choose to proportionally scale regional supplies to match base year consumption demand:

```

parameter      s0(r)   Base year supply
                d0(r)   Base year demand
                bmktol   Base year consistency (% excess demand)

s0(r) = data(r,"2000","Produce");
c0(r) = data(r,"2000","Consume");
bmktol = 100 * (sum(r,c0(r))/sum(r,s0(r)) - 1);
display bmktol;

*      Scale base year supply data to impose market
*      clearance;

alias (r,rr);

s0(r) = (s0(r)/sum(rr,s0(rr))) * sum(rr,c0(rr));

```

The static equilibrium model is then represented by the equations:

Supply ($\perp w_i$)

$$S_i = \bar{S}_i \left(\frac{w_i}{\bar{w}_i} \right)^{\eta_i} \geq \sum_j X_{ij}$$

Demand ($\perp p_j$)

$$\sum_i X_{ij} \geq D_j = \bar{D}_j \left(\frac{p_j}{\bar{p}_j} \right)^{-\sigma_j}$$

Profit ($\perp X_{ij} \geq 0$)

$$w_i + c_{ij} + t_{ij} \geq p_j$$

In setting up the model that transport costs (c_{ij}), tax rates (t_{ij}) and a single reference price (p_{USA}) are all known. The base year trade flows then are consistent with cost minimization:

$$\min \sum_{ij} X_{ij} (c_{ij} + t_{ij})$$

subject to:

$$\bar{S}_i \geq \sum_j X_{ij}$$

$$\sum_i X_{ij} \geq \bar{D}_j$$

$$X_{ij} \geq 0 \quad \forall i, j$$

Here is a summary of the steps involved in setting up a static model:

1. Move the data from its original format into something GAMS-readable. (It is a good idea to maintain a "handshake" as part of the data transfer. For example, compute row and column sums within the Excel file and then verify that these sums are identical with calculations based on the data after it has been transferred into GAMS.)
2. Write GAMS code to aggregate the number of regions to something appropriate for model development. This can be done use multi-dimensional sets (tuples) in GAMS, as illustrated in the following program:

```

set      rr      Regions in the aggregate data
/ oecd  OECD members, noecd  Countries not part of the OECD /,

r /ARG,AUS,BGD,BRA,CAN,CHE,CHL,CHN,COL,EUR,
   FSU,HKG,HUN,IDN,IND,JPN,KOR,LKA,MEX,MYS,
   NZL,PER,PHL,POL,ROW,SGP,THA,TWN,URY,USA,
   VEN,VNM/,

map(r,rr) Mapping from r to rr /
      (AUS,CAN,CHE,EUR,JPN,KOR,MEX,NZL,USA).OECD,
      (ARG,BGD,BRA,CHL,CHN,COL,FSU,HKG,HUN,IDN,IND,LKA,
       MYS,PER,PHL,POL,ROW,SGP,THA,TWN,URY,VEN,VNM).NOECD/;

*      Verify that each source region is assigned to a single disaggregate region:

parameter nmap(*) Number of regions mapping;

nmap(r) = sum(rr$rmap(r,rr), 1);
loop(r, abort$(nmap(r) ne 1) "Mapping error - r."); );
nmap(r) =0;

nmap(rr) = sum(r$rmap(r,rr), 1);
loop(rr, abort$(nmap(rr) eq 0) "Mapping error - rr"; );

*      Notice that d() is declared with a specific
*      domain prior to reading the data. This provides
*      index-checking on the first argument:

parameter      d(r,*)      Original data set
               d_(rr,*)    Aggregated data set;

$include steps.dat

*      Create the aggregate array for all the data:

alias (u,*);

d_(rr,u) = sum(r$rmap(r,rr), d(r,u));

*      Generate a new data file with the aggregated data:

file kdat /aggr.dat/; put kdat;
$libinclude gams2prm rr r
$libinclude gams2prm d_ d

```

3. Calibrate base year trade flows using linear programming. As we went through in class, this involves computing a least-cost transportation plan given supply constraints, demand constraints and transport costs. Shadow prices on demand and supply constraints can be used to infer base year prices from the LP, taking the supply price at one node as given.
4. Replicate the LP trade flows in the equilibrium model.
5. Perform a policy experiment and analyze the results.

3 Production Functions

Our initial model was based on constant elasticity supply functions. The long- and short-run elasticities of supply in this framework can be controlled by varying the value of η_i to reflect assumptions about the supply response.

3.1 A Constant Elasticity Model

A somewhat more sophisticated model would incorporate model elements which would *explain* the supply elasticity. As illustration, consider a production function which exhibits constant returns to scale in terms of two inputs: labor and capital. In the short run, capital is sector-specific and fixed, hence the function exhibits decreasing returns to scale in terms of labor inputs. We assume that the production function can be written:

$$f(K, L) = \phi (\alpha L^\rho + (1 - \alpha) K^\rho)^{1/\rho}$$

In a competitive market, market price equals marginal cost, hence:

$$p = c(w, r_K) = \frac{1}{\phi} \left(\alpha \left(\frac{w}{\alpha} \right)^{1-\sigma} + (1 - \alpha) \left(\frac{r_K}{1 - \alpha} \right)^{1-\sigma} \right)^{1/1-\sigma}$$

in which w is the wage rate, r_K is the capital rental rate, and $\sigma = 1/(1 - \rho)$ is the elasticity of substitution between labor and capital. We assume that the wage rate is given (i.e., this sector is small relative to the aggregate labor market). In the short run, capital is sector specific and fixed. By Shepard's lemma we have the following relationship between output, capital and resources:

$$y \frac{\partial c(w, r_K)}{\partial r_K} = \bar{K}$$

or:

$$\frac{y}{\phi} \left(\frac{\phi c(1 - \alpha)}{r_K} \right)^\sigma = \bar{K}$$

Following the same reasoning, we have labor demand given by:

$$\frac{y}{\phi} \left(\frac{\phi c \alpha}{w} \right)^\sigma = L$$

These demand functions characterize the model, however, if we want to compute equilibria in this framework we need to uncover the model parameters, ϕ and α . Let us assume that along a baseline steady-state growth path we can observe output (\bar{y}), employment (\bar{L}), the wage rate (\bar{w}) and the return to capital (\bar{r}_K). We may take the benchmark price of output to be unity, and the capital stock may then be inferred from the zero profit condition:

$$\bar{K} = \frac{\bar{y} - \bar{L}\bar{w}}{\bar{r}_K}$$

The demand functions for capital and labor shown above imply that:

$$\frac{\bar{L}}{\bar{K}} = \left(\frac{\alpha \bar{r}_K}{(1 - \alpha) \bar{w}} \right)^\sigma$$

or

$$\alpha = \frac{\bar{w} \bar{L}^{1/\sigma}}{\bar{w} \bar{L}^{1/\sigma} + \bar{r}_K \bar{K}^{1/\sigma}}$$

Having assigned the value of α , the value of ϕ can be inferred from the production function itself:

$$\phi = \frac{\bar{y}}{(\alpha \bar{L}^\rho + (1 - \alpha) \bar{K}^\rho)^{1/\rho}}$$

3.2 The Calibrated Form

The preceding algebra is a bit cumbersome because of the need to introduce algebraic calibration of α and ϕ , the values of which depend on the assumed elasticity of substitution σ . It can be helpful for numerical modeling to work with the *calibrated form* of the production and demand functions. In this representation the production function is written as:

$$f(K, L) = \bar{y} \left[\theta \left(\frac{L}{\bar{L}} \right)^\rho + (1 - \theta) \left(\frac{K}{\bar{K}} \right)^\rho \right]^{1/\rho}$$

in which θ is calculated as the labor value share at the benchmark point:

$$\theta = \frac{\bar{w} \bar{L}}{\bar{w} \bar{L} + \bar{r}_K \bar{K}} = \frac{\bar{w} \bar{L}}{\bar{y}}$$

Having defined the reference output price to be unity, the calibrated form of the unit cost function is written:

$$c(w, r_K) = \left[\theta \left(\frac{w}{\bar{w}} \right)^{1-\sigma} + (1 - \theta) \left(\frac{r_K}{\bar{r}_K} \right)^{1-\sigma} \right]^{1/1-\sigma}$$

The demand functions are written in calibrated form as:

$$L = \bar{L} \left(\frac{y}{\bar{y}} \right) \left(\frac{c \bar{w}}{w} \right)^\sigma$$

and

$$K = \bar{K} \left(\frac{y}{\bar{y}} \right) \left(\frac{c \bar{r}_K}{r_K} \right)^\sigma$$

3.3 Choosing σ to Match η

In a conventional application, it would be customary to assume a value of σ which, together with the base year production statistics, would define all aspects of the production function. In a dynamic model, however, it may be preferred to calibrate the value of σ to an assumed short-run elasticity of supply, i.e.

$$\eta = \left. \frac{\partial y}{\partial p} \frac{p}{y} \right|_{\bar{w}, \bar{K}}$$

In this calculation we presume that a change in the price of output results in a change in the level of employment, L , and in the return to capital r_K , while the wage rate \bar{w} and the capital stock \bar{K} remain fixed.

When the wage departs from its benchmark value, assuming competitive supply (i.e., $c(r_K, w) = p$), the capital constraint can be inverted to obtain an explicit expression for the return in terms of the output price and output quantity:

$$\frac{r_k}{\bar{r}_K} = p \left(\frac{y}{\bar{y}} \right)^{1/\sigma}$$

Substituting back into the cost function, we have

$$p^{1-\sigma} = \theta p^{1-\sigma} \left(\frac{y}{\bar{y}} \right)^{\frac{1-\sigma}{\sigma}} + (1-\theta) w^{1-\sigma}$$

or

$$\frac{y}{\bar{y}} = \theta^{\frac{1}{\sigma-1}} \left[1 - (1-\theta) \left(\frac{w}{p} \right)^{1-\sigma} \right]^{\frac{\sigma}{1-\sigma}}$$

Differentiating this expression with respect to the relative price of output, and setting all prices equal to unit, we have:

$$\eta = \frac{\sigma(1-\theta)}{\theta}$$

This equation can be used in a variety of ways to calibrate the supply function. One approach would be to choose the value share of the fixed factor θ to match the base year profits, and then assign the elasticity according to:

$$\sigma = \eta \frac{\theta}{(1-\theta)}$$

Alternatively, one could take the elasticity of substitution as given and the specific factor value share accordingly:

$$\theta = \frac{\sigma}{\sigma + \eta}$$

3.4 A Three Factor Model

In a dynamic model it may be helpful to introduce two notions of the elasticity of supply: short-run and long-run. A simple way to introduce this distinction into a numerical model is to work with a three factor production function:

$$y = f(L, K, R)$$

where L is labor, a production factor which is variable in both the short and long run, K is capital, a quasi-fixed production factor which is variable in the long run but fixed in the short run, and R is a sector-specific resource which is fixed in both the short and long run.

Following the approach outlined in the previous section, we can write the long-run production function as:

$$y = \bar{y} \left[\theta_L \left(\frac{L}{\bar{L}} \right)^\rho + \theta_K \left(\frac{K}{\bar{K}} \right)^\rho + (1 - \theta_L - \theta_K) \right]^{1/\rho}$$

The R input does not appear in the calibrated production function because we have assumed that $R = \bar{R}$. In the short-run model, we have:

$$y = \bar{y} \left[\theta_L \left(\frac{L}{\bar{L}} \right)^\rho + (1 - \theta_L) \right]^{1/\rho}$$

If the short-run elasticity of supply is given by η_S , and the labor value share (θ_L) is given in the benchmark data, we can then calibrate the elasticity of substitution to match these inputs:

$$\sigma = \eta_S \frac{\theta_L}{(1 - \theta_L)}$$

Let us assume that while the labor value share is observed (employment statistics are commonly available for many sectors and regions), but the allocation of the remaining value added is not known with certainty. We can take advantage of this uncertainty to calibrate the long-run supply response by choosing the capital value as:

$$\theta_K = \frac{\sigma}{\sigma + \eta_L} - \theta_L$$

given the value of σ previously calibrated. (It can be shown that when $\eta_L > \eta_S$, we find that $\theta_K < 1 - \theta_L$.)

4 Quadratic Adjustment Cost

By convention we model the capital stock (K_t) as subject to a constant geometric depreciation rate. The capital stock may be increased by *net* investment (I_t):

$$K_{t+1} = (1 - \delta)K_t + I_t$$

Following Uzawa, we assume that capital installation costs depend on the rate of gross investment relative to the existing capital stock. Given the level of investment, the cost of new capital decreases when the capital stock increases and vice versa. The installation cost function relating net to *gross* investment is given by

$$J_t = I_t \left(1 + \phi \frac{I_t}{2K_t} \right)$$

where ϕ is the adjustment cost parameter.

When $\phi = 0$, there are no adjustment costs, and the model reduces to the neoclassical (Ramsey) model. When ϕ is large, rapid changes in the capital stock are costly and the speed of adjustment is reduced when installation costs increase. It follows from this specification that net investment is included in the intertemporal market clearance condition for capital and that gross investment is included in the market clearance condition for output.

Along a balanced growth path with growth rate g and depreciation rate δ , $I = (g + \delta)K$, and the marginal cost of investment equals $1 + \phi(g + \delta)$, and with a one period investment lag the base year capital price is

$$p_K = (1 + r) [1 + \phi(g + \delta)].$$

As a result of adjustment costs, a marginal increase in the capital stock both increases the supply of productive capital services and decreases the cost of new investment. We can evaluate the marginal return associated decreased investment cost by computing:

$$\frac{\partial J}{\partial K} = -\frac{\phi I^2}{2K^2}$$

The baseline investment cost premium for capital is therefore given by:

$$\rho = \frac{\phi(g + \delta)^2}{2}$$

With a competitive market for capital services, the value of capital at the start of a period equals the return to productive services in that period (r_K), the premium associated with investment (ρ) and the salvage value of capital in the subsequent period:

$$p_K = r_K + \rho + (1 - \delta) \frac{p_K}{1 + r}$$

Hence,

$$r_K = \frac{r + \delta}{1 + r} p_K - \rho$$

which reduces to:

$$r_K = r + \delta + \phi(\delta + g) \left[r + \frac{\delta - g}{2} \right]$$

The rental price of capital then includes three terms: (i) the time cost (r), (ii) the depreciation cost (δ), and (iii) the adjustment premium ($\phi(\delta + g) \left[r + \frac{\delta - g}{2} \right]$).

5 A Dynamic Optimization Model

I've coded up a single production dynamic optimization model which illustrates both the calibration of short- and long-run supply elasticities and the Uzawa adjustment cost representation of investment. This GAMS program represents the producer problem both as a nonlinear optimization problem and as a mixed complementarity model. There are seven sections to the program, as follows.

5.1 Set definitions and input assumptions

```
set      t      Time period in the model (year) /2002*2050/,
        td(t)   Decades          /2010,2020,2030,2040,2050/;

scalar  lvs      Base year labor value share    /0.4/,
        phi      Speed of adjustment parameter /0.3/,
        r        Interest rate                 /0.05/,
        g        Growth rate                   /0.02/,
        delta    Capital depreciation rate      /0.07/,
        eta_S    Short-run elasticity of supply /0.5/,
        eta_L    Long-run elasticity of supply  /1.5/;
```

5.2 Calibration

```
sets    tf(t)    First period of the model,
        tl(t)    Last period of the model;

tf(t) = yes$(ord(t) eq 1);
tl(t) = yes$(ord(t) eq card(t));

*      Calibrated parameters for the production function:

parameter
    price(t)    Output price,
    sigma        Elasticity of substitution,
    rho          Primal elasticity parameter,
    rk0          Baseline rental price of capital,
    kvs          Capital value share,
    k0           Base year capital stock,
    i0           Base year investment,
    rvs          Resource value share,
    rs(t)        Resource stock (index),
    p(t)         Steady-state price index;

*      Rental rate on capital is defined by the
*      capital adjustment cost model and the
*      assumption of a steady-state growth path:

rk0 = r + delta + phi * (delta+g) * (r+(delta-g)/2);

*      Steady-state prices decline with the
*      interest rate:

p(t) = 1/power(1+r,ord(t)-1);

*      Steady-state quantities increase with the
*      growth rate:

rs(t) = power(1+g,ord(t)-1);

*      Use the labor value share and the short-run
*      elasticity of supply to calibrate the elasticity
*      of substitution:

sigma = (1-lvs)/lvs * eta_S;
rho = 1 - 1/sigma;

*      Use the elasticity of substitution and the long-run
*      elasticity of supply to calibrate the resource value
*      share:

rvs = sigma/(eta_L + sigma);

*      Capital value share is then determined as a residual:
```

```

kvs      = 1 - rvs - lvs;

*      Capital stock is defined by the capital earnings
*      (which equal the capital value share given the
*      value of output scaled to unity):

k0       = kvs/rk0;

*      Net investment is consistent with steady-state
*      growth:

i0       = k0 * (g+delta);

display sigma, kvs, rvs;

```

5.3 The nonlinear programming model

```

variables      PI      Present value of profit

positive
variables
      Y(t)      Production in period t
      I(t)      Net Investment
      J(t)      Gross Investment
      L(t)      Employment
      K(t)      Capital
      KT        Terminal capital stock;

equations profit, jdef, terminv, capital, termcap, output;

*      Objective is to maximize the present value of
*      profits over the model horizon and the salvage
*      value of capital in the final period:

profit..      pi =e=
              sum(t, (price(t)*y(t)-lvs*L(t)-J(t)) * p(t))
              + sum(tl, p(tl)*(1+phi*(g+delta))*kt);

*      Gross investment equals net investment plus
*      adjustment costs:

jdef(t)..      J(t) =e= I(t) * (1 + phi * I(t)/(2*K(t)));

*      To approximate an infinite horizon, we require
*      investment in the terminal period equals
*      the steady-state value:

terminv..      sum(tl, I(tl) - (g+delta)*K(tl)) =e= 0;

*      Capital evolves through depreciation and net
*      investment:

capital(t)..    (1-delta)*K(t-1) + I(t-1) + k0$tf(t) =e= K(t);

termcap..      sum(tl, (1-delta) * K(tl) + I(tl)) =e= kt;

output(t)..     y(t) =e= (lvs*L(t)**rho
                        + kvs*(K(t)/k0)**rho
                        + rvs*rs(t)**rho)**(1/rho);

k.l(t) = rs(t) * k0;
I.l(t) = rs(t) * i0;
J.l(t) = I.l(t) * (1 + phi * I.l(t)/(2*K.l(t)));
L.l(t) = rs(t);
y.l(t) = rs(t);
kt.l = (1+g)**card(t) * k0;
pi.l = sum(t,(y.l(t)-lvs*L.l(t)-J.l(t))*p(t))+sum(tf,p(tf)*kt.l);

model dynamic /all/;

*      Insert some bounds to prevent bad function calls:

k.lo(t) = 0.01*rs(t);
l.lo(t) = 0.01*rs(t);
k.fx(tf) = k0;

price(t) = 1;

solve dynamic using nlp maximizing pi;

```


5.4 Check of steady-state calibration

```

parameter          summary          Summary report (% change from steady state);

*      Install deviations from the steady-state path.

*      All of these values should be zero.

*      Add "eps" so that everything gets plotted, otherwise
*      zeros are ignored and nothing would be plotted:

summary(t,"y") = 100 * (y.l(t)/rs(t)-1) + eps;
summary(t,"k") = 100 * (k.l(t)/(k0*rs(t))-1) + eps;
summary(t,"i") = 100 * (I.l(t)/(i0*rs(t))-1) + eps;
summary(t,"L") = 100 * (L.l(t)/rs(t)-1) + eps;

$setglobal domain t
$setglobal labels td
$libinclude plot summary

```

5.5 The complementarity model

```

variables          pk(t)   Purchase price of capital,
                   rk(t)   Rental price of capital,
                   rr(t)   Rental price of the fixed factor,
                   pkT     Terminal price of capital;

equations          profit_i(t)   Investment profit
                   profit_k(t)   Capital profit
                   profit_y(t)   Production profit
                   capstock(t)   Capital services market
                   resource(t)   Resource constraint;

profit_i(t)..     p(t) * (1 + phi * I(t)/K(t))
                  =g= pk(t+1) + pkT$t1(t);

profit_k(t)..     pk(t) =e= rk(t)

                  + (phi/2) * sqrt(I(t)/K(t)) * p(t)

                  + (1-delta)*(pk(t+1)+pkT$t1(t));

capstock(t)..     K(t) =e= y(t) * k0 * (rk0*price(t)*p(t)/rK(t))**sigma;

resource(t)..     rs(t) =e= y(t) * (price(t)*p(t)/rr(t))**sigma;

profit_y(t)..     ( lvs
                  + kvs * (rk(t)/(p(t)*rk0))**(1-sigma)
                  + rvs * (rr(t)/p(t))**(1-sigma) )**(1/(1-sigma))
                  =e= price(t);

model dynamic_mcp /profit_y.y, profit_k.k, profit_i.I,
                  capital.pk, termcap.pkT, terminv.kt,
                  resource.rr, capstock.rk/;

*      Install steady-state values for prices:

rr.l(t) = p(t);
rk.l(t) = p(t)*rk0;
pk.l(t) = p(t)*(1+r)*(1+phi*(g+delta));
pkT.l = sum(tl, pk.l(tl)/(1+r));

parameter pricechk   Cross check of steady-state prices;

pricechk(t,"rr") = rr.l(t) - (output.m(t) * (y.l(t)/rs(t))**(1-rho));
pricechk(t,"rk") = rk.l(t) - (kvs * k0**(-rho) * output.m(t) * (y.l(t)/k.l(t))**(1-rho));
pricechk(t,"pk") = pk.l(t) + capital.m(t);
pricechk(tf,"pk") = pk.l(tf) - k.m(tf);
pricechk(t,"pkT") = pkT.l - sum(tl,p(tl)*(1+phi*(g+delta)));
display pricechk;

*      Replicate the baseline steady-state growth path:

dynamic_mcp.iterlim = 0;
solve dynamic_mcp using mcp;

```

5.6 Checking short- and long-run supply elasticities

```
*      Perturb the output prices and trace the short-
*      and long-run supply response:

price(t) = 1.01;

solve dynamic using nlp maximizing pi;

parameter          summary Optimal response;
summary(t,"RR") = 100 * (
    output.m(t) * (y.l(t)/rs(t))**(1-rho) / p(t) - 1);
summary(t,"RK") = 100 * (
    kvs*k0**(-rho)*output.m(t)*(y.l(t)/k.l(t))**(1-rho)
    / (p(t)*rk0) - 1);
summary(t,"Y") = 100 * (y.l(t)/rs(t)-1) + eps;
summary(t,"K") = 100 * (k.l(t)/(k0*rs(t))-1) + eps;
summary(t,"I") = 100 * (i.l(t)/(i0*rs(t))-1) + eps;

*      Plot the percentage change in output:

$libinclude plot summary
display summary;
```

5.7 Verify consistency of the NLP and MCP models

```
*      Verify that the NLP and MCP models are
*      consistent by checking the NLP solution:

rr.l(t) = output.m(t) * (y.l(t)/rs(t))**(1-rho);
rk.l(t) = kvs * k0**(-rho) * output.m(t) * (y.l(t)/k.l(t))**(1-rho);
pk.l(t) = -capital.m(t);
pk.l(tf) = k.m(tf);
pkT.l = sum(tl,p(tl)*(1+phi*(g+delta)));

dynamic_mcp.iterlim = 0;
solve dynamic_mcp using mcp;

display "Tolerance of the MCP solution is:",dynamic_mcp.OBJVAL;
```

6 Scenario Analysis

6.1 run.bat

```
@echo off
if exist output.sol del output.sol

echo $setglobal phi 1 >nlp.ini
call gams nlp

echo $setglobal phi 2 >nlp.ini
call gams nlp

echo $setglobal phi 3 >nlp.ini
call gams nlp
```

6.2 nlp.gms

```
$title Single Producer Dynamic Production Model

$if exist nlp.ini $include nlp.ini

$if not setglobal phi $setglobal phi 3

set t Time period in the model (year) /2002*2050/,
ts(t) Time period of the price shock /2002*2010/,
td(t) Decades /2010,2020,2030,2040,2050/;

scalar lvs Base year labor value share /0.4/,
phi Speed of adjustment parameter /0.%phi%/;
```

Figure 1: Consistency Check for Steady-state Calibration





Figure 2: Dynamic Response of a 1% Price Shock



```

r Interest rate /0.05/,
g Growth rate /0.02/,
delta Capital depreciation rate /0.07/,
eta_S Short-run elasticity of supply /0.5/,
eta_L Long-run elasticity of supply /1.5/;

sets tf(t) First period of the model,
tl(t) Last period of the model;

tf(t) = yes$(ord(t) eq 1);
tl(t) = yes$(ord(t) eq card(t));

parameter
price(t) Output price,
sigma Elasticity of substitution,
rho Primal elasticity parameter,
rk0 Baseline rental price of capital,
kvs Capital value share,
k0 Base year capital stock,
i0 Base year investment,
rvs Resource value share,
rs(t) Resource stock (index),
p(t) Steady-state price index;

rk0 = r + delta + phi * (delta+g) * (r+(delta-g)/2);
p(t) = 1/power(1+r,ord(t)-1);
rs(t) = power(1+g,ord(t)-1);
sigma = (1-lvs)/lvs * eta_S;
rho = 1 - 1/sigma;
rvs = sigma/(eta_L + sigma);
kvs = 1 - rvs - lvs;
k0 = kvs/rk0;
i0 = k0 * (g+delta);

variables PI Present value of profit
positive
variables Y(t) Production in period t
I(t) Net Investment
J(t) Gross Investment
L(t) Employment
K(t) Capital
KT Terminal capital stock;

equations profit, jdef, terminv, capital, termcap, output;

profit.. pi =e=
sum(t, (price(t)*y(t)-lvs*L(t)-J(t)) * p(t))
+ sum(tl, p(tl)*(1+phi*(g+delta))*kt);

jdef(t).. J(t) =e= I(t) * (1 + phi * I(t)/(2*K(t)));

terminv.. sum(tl, I(tl) - (g+delta)*K(tl)) =e= 0;

capital(t).. (1-delta)*K(t-1) + I(t-1) + k0$tf(t) =e= K(t);

termcap.. sum(tl, (1-delta) * K(tl) + I(tl)) =e= kt;

output(t).. y(t) =e= (lvs*L(t)**rho + kvs*(K(t)/k0)**rho + rvs*rs(t)**rho)**(1/rho);

k.l(t) = rs(t) * k0;
I.l(t) = rs(t) * i0;
J.l(t) = I.l(t) * (1 + phi * I.l(t)/(2*K.l(t)));
L.l(t) = rs(t);
y.l(t) = rs(t);
kt.l = (1+g)**card(t) * k0;
pi.l = sum(t,(y.l(t)-lvs*L.l(t)-J.l(t))*p(t))+sum(tf,p(tf)*kt.l);

model dynamic /all/;

k.lo(t) = 0.01*rs(t); l.lo(t) = 0.01*rs(t); k.fx(tf) = k0; price(t) = 1;

solve dynamic using nlp maximizing pi;

price(ts) = 1.01;

solve dynamic using nlp maximizing pi;

parameter ypct Output response to price change;

ypct(t,"%phi%") = 100 * (y.l(t)/rs(t)-1) + eps;

file kout /output.sol/; kout.ap=1; put kout;

$libinclude gams2txt ypct

```

6.3 output.sol

```
==>gams2txt ypct
* Called from E:\COURSES\6433\NLP.GMS -- line 359
* 03/17/02 23:00:39
2002.1      4.9772312417002E-01
2003.1      1.0803120954707E+00
2004.1      1.3231263482286E+00
2005.1      1.4227954386266E+00
2006.1      1.4605473097837E+00
2007.1      1.4670618744718E+00
2008.1      1.4470035834035E+00
2009.1      1.3821975484625E+00
...
```

6.4 report.gms

```
set t /2002*2050/
tf(t), tl(t),
td(t) /2005,2010,2015,2020/,
tearly(t) /2002*2020/

parameter ypct(t,*) /
$include output.sol
/;

$setglobal domain tearly
$setglobal labels td
$libinclude plot ypct
```

Figure 3: Dynamic Response of a Temporary Price Shock

